Design Project

ECE4445 – Audio Engineering

November 27th, 2006

Franklin Falcon

901-95-6137

ffalcon@gatech.edu

Juan O'Connell

570-575-8161

gtg131v@mail.gatech.edu

In this design project, various simulations were performed. The simulations are:

1. Loudspeaker mounted on infinite baffle.
2. Loudspeaker mounted inside a closed box.
3. Loudspeaker with matching network mounted inside a closed box.
4. Loudspeaker with matching network and 2^{nd} order crossover network mounted inside a closed box.
5. Loudspeaker with matching network and 3^{rd} order crossover network mounted inside a closed box.
6. Loudspeaker mounted inside a vented box.

All the simulations were performed using PSPICE. The SPICE circuits and netlists are attached at the end of the report.

$Zoff := READPRN("off.txt")$ $Zon := READPRN("on.txt")$

$Zoff =$

	0	1	2
0	9.994	8.258	40.046
1	10.244	8.39	40.636
2	10.517	8.555	40.839
3	10.792	8.704	41.537
4	11.067	8.876	41.815
5	11.34	9.034	42.706
6	11.64	9.225	43.056
7	11.939	9.436	42.956
8	12.239	9.61	44.423
9	12.538	9.815	45.014
10	12.873	10.048	45.625
11	13.222	10.31	45.932
12	13.546	10.558	46.356
13	13.895	10.827	46.886
14	14.245	11.117	47.236
15	14.619	11.432	47.697

$Zon =$

	0	1	2
0	9.994	6.021	10.063
1	10.244	6.03	10.058
2	10.518	6.039	10.25
3	10.792	6.049	10.568
4	11.067	6.058	10.728
5	11.341	6.066	11.26
6	11.641	6.077	11.345
7	11.94	6.088	11.499
8	12.239	6.097	11.906
9	12.538	6.108	12.123
10	12.873	6.12	12.43
11	13.222	6.135	12.743
12	13.546	6.147	13.018
13	13.896	6.16	13.357
14	14.245	6.174	13.608
15	14.619	6.191	13.935

$R_E := 5.33$ DC resistance $V_T := 1.56$ Test box volume

$N := 300$ Number of data points minus 1 $Z_p(x,y) := \frac{x \cdot y}{x + y}$ Parallel combinaton

From the Zoff array:

$f_S := 30.482$ $R_{ES} := 43.614 - R_E$ $R_{ES} = 38.284$ $R_1 := \sqrt{R_E \cdot (R_E + R_{ES})}$

$R_1 = 15.247$ $f_1 := 18.36$ $f_2 := 50.582$ $\sqrt{f_1 \cdot f_2} = 30.474$

$Q_{MS} := \frac{f_S}{f_2 - f_1} \cdot \sqrt{\frac{R_E + R_{ES}}{R_E}}$ $Q_{MS} = 2.706$ $Q_{ES} := \frac{R_E}{R_{ES}} \cdot Q_{MS}$ $Q_{ES} = 0.377$

$Q_{TS} := \frac{R_E}{R_E + R_{ES}} \cdot Q_{MS}$ $Q_{TS} = 0.331$

$$n_e := \frac{2}{\pi}\cdot\arg\left(Zoff_{250,1}\cdot\cos\left(Zoff_{250,2}\cdot\frac{\pi}{180}\right) - R_E + j\cdot Zoff_{250,1}\cdot\sin\left(Zoff_{250,2}\cdot\frac{\pi}{180}\right)\right) \qquad n_e = 0.65$$

$$L_e := \frac{\sqrt{\left(Zoff_{250,1}\cdot\cos\left(Zoff_{250,2}\cdot\frac{\pi}{180}\right) - R_E\right)^2 + \left(Zoff_{250,1}\cdot\sin\left(Zoff_{250,2}\cdot\frac{\pi}{180}\right)\right)^2}}{\left(2\cdot\pi\cdot Zoff_{250,0}\right)^{n_e}} \qquad L_e = 0.03$$

$n := 0 .. N \qquad f_n := 10\cdot\left(\frac{20000}{10}\right)^{\frac{n}{N}}$ Frequency range variable for calculating Zvc(f)

$L_E := 13\cdot 10^{-3}$ "Tweaked" value of parallel lossless inductor in Allen Robinson's model

$$Zvc(f) := R_E + Z_p\left[j\cdot 2\cdot\pi\cdot f\cdot L_E, L_e\cdot(j\cdot 2\cdot\pi\cdot f)^{n_e}\right] + R_{ES}\cdot\frac{\frac{1}{Q_{MS}}\cdot\left(\frac{j\cdot f}{f_S}\right)}{1 - \left(\frac{f}{f_S}\right)^2 + \frac{1}{Q_{MS}}\cdot\left(\frac{j\cdot f}{f_S}\right)}$$

From the Zon array:

$f_{CT} := 75.889 \qquad R_{ECT} := 34.353 - R_E \qquad R_{ECT} = 29.023 \qquad R_1 := \sqrt{R_E\cdot(R_E + R_{ECT})}$

$R_1 = 13.531 \qquad f_{1T} := 55.988 \qquad f_{2T} := 95.33 \qquad \sqrt{f_{1T}\cdot f_{2T}} = 73.057$

$$Q_{MCT} := \frac{f_{CT}}{f_{2T} - f_{1T}}\cdot\sqrt{\frac{R_E + R_{ECT}}{R_E}} \qquad Q_{MCT} = 4.897$$

$$Q_{ECT} := \frac{R_E}{R_{ECT}}\cdot Q_{MCT} \qquad Q_{ECT} = 0.899 \qquad Q_{TCT} := \frac{R_E}{R_E + R_{ECT}}\cdot Q_{MCT} \qquad Q_{TCT} = 0.76$$

$$V_{AS} := V_T\cdot\left(\frac{f_{CT}}{f_S}\cdot\frac{Q_{ECT}}{Q_{ES}} - 1\right) \qquad V_{AS} = 7.711$$


Impedance Magnitude
100
10
1
Impedance [ohms]
Measured
Calculated
1
10
100
1 ·10^3
1 ·10^4
1 ·10^5
Frequency [Hz]


Impedance Phase
60
40
20
0
−20
−40
−60
Phase [deg.]
Measured
Calculated
1
10
100
1 ·10^3
1 ·10^4
1 ·10^5
Frequency [Hz]

Infinite Baffle:

In this section, parameters are calculated to construct the electrical circuits which simulate the speaker mounted on an infinite baffle. From these electrical circuits, it is possible to obtain the input impedance, the sound pressure level produced by the loudspeaker, and the diaphragm peak displacement. The equations used for the calculations are shown below. Also, the graphics show the parameters measured from the SPICE simulation.

$Q_{MS} = 2.706$ $\quad Q_{ES} = 0.377$ $\quad Q_{TS} = 0.331$

$\rho_o := 1.18$ $\quad c := 345$ $\quad a := 0.12$ $\quad V_{AS} := 0.218$ ---> in m^3

$f_S = 30.482$ $\quad V_{AS} = 0.218$ $\quad S_D := \pi a^2$ $\quad S_D = 0.045$

$$C_{MS} := \frac{V_{AS}}{\rho_o \cdot c^2 \cdot S_D^{\,2}} \qquad C_{MS} = 7.584 \times 10^{-4}$$

$$M_{MS} := \frac{1}{(2\pi \cdot f_S)^2 \cdot C_{MS}} \qquad M_{MS} = 0.036$$

$$R_{MS} := \frac{1}{Q_{MS}} \cdot \sqrt{\frac{M_{MS}}{C_{MS}}} \qquad R_{MS} = 2.544$$

$$Bl := \sqrt{\frac{R_E}{Q_{ES}} \cdot \sqrt{\frac{M_{MS}}{C_{MS}}}} \qquad Bl = 9.869$$

$$M_{MD} := M_{MS} - 2 \cdot \frac{8 \cdot \rho_o}{3\pi^2 \cdot (a)} \cdot S_D^{\,2} \qquad M_{MD} = 0.025$$

$$M_{A1} := \frac{8 \cdot \rho_o}{3 \cdot \pi^2 \cdot (a)} \qquad M_{A1} = 2.657$$

$$R_{A2} := \frac{\rho_o \cdot c}{\pi \cdot a^2} \qquad R_{A2} = 8.999 \times 10^3$$

$$R_{A1} := \frac{128 \cdot \rho_o \cdot c}{9 \cdot \pi^3 \cdot a^2} - R_{A2} \qquad R_{A1} = 3.969 \times 10^3$$

$$C_{A1} := \frac{5.94 \cdot a^3}{\rho_o \cdot c^2} \qquad C_{A1} = 7.308 \times 10^{-8}$$

SpcIBZin := READPRN("SpiceIBZin.txt")



SpcIBSPL := READPRN("SpiceIBSPL.txt")



SpcIBXpp := READPRN("SpiceIBXpp.txt")



Closed Box:

In this section, parameters are calculated to build the circuits that model the loudspeaker mounted inside a closed box. The first three graphs show the results of the SPICE measurements. A matching network was designed with the purpose of making the input impedance behave like if it was only the voice coil resistor RE, the graph shown under "Matching Network Design" shows the electrical input impedance before and after adding the matching network. Also, 2nd and 3rd order crossover networks were designed with the purpose of only allowing low frequencies to reach the woofer. The measured sound pressure levels without crossover network, with a 2nd order network, and with a 3rd order network, are shown in the graph under "Crossover Network Design".

$$\alpha := \left(\frac{Q_{ECT}}{Q_{ES}}\right)^2 - 1 \qquad \alpha = 4.698$$

$$V_{AB} := \frac{V_{AS}}{\alpha} \qquad V_{AB} = 0.046$$

$$C_{AB} := \frac{V_{AB}}{\rho_o \cdot c^2} \qquad C_{AB} = 3.304 \times 10^{-7}$$

$$R_{AB} := \left(\sqrt{1 + \alpha} \cdot \frac{Q_{MS}}{Q_{MCT}}\right) \cdot \frac{R_{MS}}{S_D^{\,2}} \qquad R_{AB} = 1.64 \times 10^3$$

$$M_{MC} := \left(\frac{Bl^2 \cdot Q_{ECT}}{\pi \cdot a^2 \cdot c \cdot R_E}\right)^2 \cdot \frac{V_{AS}}{\rho_o \cdot (1 + \alpha)} \qquad M_{MC} = 0.036$$

$$R_{MS} := \frac{1}{Q_{MS}} \cdot \sqrt{\frac{M_{MC}}{C_{MS}}} \qquad R_{MS} = 2.544$$

$$M_{AB} := \frac{M_{MC} - M_{MD}}{\pi^2 \cdot a^4} - M_{A1} \qquad M_{AB} = 2.657$$

$$R_{AL} := \frac{1}{2\pi(.1) \cdot C_{AB}} \qquad R_{AL} = 4.818 \times 10^6$$

SpcCBZin := READPRN("SpiceCBZin.txt")

SpcCBSPL := READPRN("SpiceCBSPL.txt")

SpcCBXpp := READPRN("SpiceCBXpp.txt")


Electrical Input Impedance
Impedance [Ohms]
0
20
40
10
100
1·10^3
Frequency [Hz]


On-Axis SPL
SPL [dB]
60
80
10
100
1·10^3
Frequency [Hz]


Peak to Peak Diaphragm Displacement
Displacement [m]
0.01
1·10^−3
1·10^−4
1·10^−5
1·10^−6
1·10^−7
1·10^−8
10
100
1·10^3
Frequency [Hz]

Matching Network Design:

$f_c := f_S \cdot \sqrt{1+\alpha}$ $\quad f_c = 72.764$ $\quad f_{1mn} := 2000$ $\quad f_{2mn} := 20 \cdot 10^3$ $\quad L_e = 0.03$ $\quad n_e = 0.65$

$R_E = 5.33$ $\quad Q_{MCT} = 4.897$ $\quad Q_{ECT} = 0.899$ $\quad Q_{EC} := Q_{ES} \cdot \sqrt{1+\alpha}$

$R_{1.mn} := R_E$

$$C_{1.mn} := \frac{L_e}{(2\pi)^{(1-n_e)} \cdot R_E^{\,2} \cdot \left[f_{1mn}^{\,n_e} \cdot f_{2mn}^{\,(2+n_e)} \right]^{\frac{(1-n_e)}{2(1+n_e)}}}$$

$$C_{2.mn} := \frac{L_e}{(2\pi)^{1-n_e} \cdot R_E^{\,2} \cdot \left[f_{1mn}^{\,(2+n_e)} \cdot f_{2mn}^{\,n_e} \right]^{\frac{1-n_e}{2(1+n_e)}}} - C_{1.mn}$$

$$R_{2.mn} := \frac{1}{2\pi f_{1mn}^{\frac{1}{(1+n_e)}} \cdot f_{2mn}^{\frac{n_e}{(1+n_e)}} \cdot C_{2.mn}}$$

$$R_{3.mn} := R_E \cdot \left(1 + \frac{Q_{ECT}}{Q_{MCT}} \right) \qquad L_{1.mn} := \frac{R_E \cdot Q_{ECT}}{2\pi f_c} \qquad C_{3.mn} := \frac{1}{2\pi f_c \cdot R_E \cdot Q_{ECT}}$$

$R_{1.mn} = 5.33$ $\quad C_{1.mn} = 2.054 \times 10^{-5}$ $\quad C_{2.mn} = 1.293 \times 10^{-5}$ $\quad R_{2.mn} = 2.484$

$R_{3.mn} = 6.309$ $\quad L_{1.mn} = 0.01$ $\quad C_{3.mn} = 4.563 \times 10^{-4}$

SpcCBwMNZin := READPRN("SpiceCBwMNZin.txt")

Electrical Input Impedance



Crossover Network Design:

. 2nd Order Network:

Assuming $Q_w := 0.5$ $f_w := 800$ $R_E = 5.33$

$$L_w := \frac{R_E}{2\pi f_w \cdot Q_w} \qquad C_w := \frac{Q_w}{2\pi f_w \cdot R_E} \qquad L_w = 2.121 \times 10^{-3} \qquad C_w = 1.866 \times 10^{-5}$$

. 3rd Order Network:

$f_c = 72.764$ $f_w = 800$

$$L_2 := \frac{R_E}{4\pi f_w} \qquad L_1 := 3L_2 \qquad C_3 := \frac{2}{3\pi f_w \cdot R_E}$$

$L_1 = 1.591 \times 10^{-3}$ $L_2 = 5.302 \times 10^{-4}$ $C_3 = 4.977 \times 10^{-5}$

SpcCBw2COSPL := READPRN("SpiceCBwMN2COSPL.txt")

SpcCBw3COSPL := READPRN("SpiceCBwMN3COSPL.txt")



Vented Box:

In this section, parameters are calculated to build the circuits that will simulate the behavior of the loudspeaker mounted inside a vented box. The electrical input impedance was measured, along with the sound presure level and the peak displacement. The graph labeled "On-Axis SPL" shows the sound pressure levels produced by the diaphragm and the air in the vent, along with the total SPL. The plot labeled "Peak-to-Peak Displacement" shows the displacement of the diaphragm as well as the displacement of the air in the port.

$d_w := 0.305$ --> 12 in = woofer frame diameter

$a_w := 0.12$ --> 12 cm = piston radius

$a_p := 0.04$ --> 4 cm = port radius

woofer-port spacing assumed to be 1.6 times the woofer frame radius, then:

$d_1 := 0.8d_w$ $\quad d_1 = 0.244$

$S_w := \pi a_w^2$ $\quad S_w = 0.045$

$B := 0.65$. --> Assumed Value (p.114)

$M_{AB} := \dfrac{B \cdot \rho_o}{\pi \cdot a_w}$ $\quad M_{AB} = 2.035$ Old Mab was 2.657

$C_{AB} = 3.304 \times 10^{-7}$

$Q_{TS} = 0.331$ $\quad Q_L := 7$

With assumed QL=7 and Alignment Chart on p.138:

$h := 1.1998$ $\quad f_B := h \cdot f_S$ $\quad f_B = 36.572$

$$R_{AL} := \frac{Q_L}{2\pi f_B \cdot C_{AB}} \qquad R_{AL} = 9.221 \times 10^4$$

$$M_{A1P} := \frac{8\rho_o}{3\pi^2 \cdot a_p} \qquad M_{A1P} = 7.971$$

$$M_{AP} := \frac{1}{(2\pi f_B)^2 \cdot C_{AB}} - M_{A1P} \qquad M_{AP} = 49.355$$

$$R_{A2P} := \frac{\rho_o \cdot c}{\pi a_p^{\ 2}} \qquad R_{A2P} = 8.099 \times 10^{4}$$

$$R_{A1P} := \frac{128\rho_o \cdot c}{9\pi^3 \cdot a_p^{\ 2}} - R_{A2P} \qquad R_{A1P} = 3.572 \times 10^{4}$$

$$C_{A1P} := \frac{5.94 a_p^{\ 3}}{\rho_o \cdot c^2} \qquad C_{A1P} = 2.707 \times 10^{-9}$$

$$k_p := \frac{3\pi a_p}{16 d_1} \qquad k_p = 0.097$$

$$S_P := \pi a_p^{\ 2} \qquad S_P = 5.027 \times 10^{-3}$$

$$M_{A1W} := \frac{8\rho_o}{3\pi^2 \cdot a_w} \qquad M_{A1W} = 2.657$$

$$R_{A2W} := \frac{\rho_o \cdot c}{\pi a_w^{\ 2}} \qquad R_{A2W} = 8.999 \times 10^{3}$$

$$R_{A1W} := \frac{128\rho_o \cdot c}{9\pi^3 \cdot a_w^{\ 2}} - R_{A2W} \qquad R_{A1W} = 3.969 \times 10^{3}$$

$$C_{A1W} := \frac{5.94 a_w^{\ 3}}{\rho_o \cdot c^2} \qquad C_{A1W} = 7.308 \times 10^{-8}$$

$$k_w := \frac{3\pi a_w}{16 d_1} \qquad k_w = 0.29$$

SpcVBZin := READPRN("SpiceVBZin.txt")   SpcVBSPLv := READPRN("SpiceVBSPLv.txt")

SpcVBSPLw := READPRN("SpiceVBSPLw.txt")   SpcVBSPL := READPRN("SpiceVBSPL.txt")

SpcVBXppD := READPRN("SpiceVBXppD.txt")   SpcVBXppP := READPRN("SpiceVBXppP.txt")


Electrical Input Impedance
Impedance [Ohms]
40
20
0
10
100
1·10^3
1·10^4
Frequency [Hz]


On-Axis SPL
SPL [dB]
80
60
Total
Woofer
Vent
10
100
1·10^3
1·10^4
Frequency [Hz]


Peak to Peak Displacement
Displacement [m]
0.01
1·10^−3
1·10^−4
1·10^−5
1·10^−6
Diaphragm
Port Air
10
100
1·10^3
1·10^4
Frequency [Hz]

Summary and Conclusions

This report shows the result of simulating the behavior of a loudspeaker under different operating conditions. In the case of the closed box simulation, a matching network was placed in parallel with the voice coil. The purpose of this network is to cause the voice coil impedance to behave as if it were purely resistive, i.e. to cancel the peak at the resonant frequency as well as the high frequency rise due to the voice coil inductance. Furthermore, 2^{nd} and 3^{rd} order crossover networks were placed at the input of the voice coil circuit, i.e. before the matching network. The purpose of these networks is to allow only low frequencies to reach the woofer. The plots of the sound pressure level when using the crossover networks show how the slope of the magnitude Bode plot changes from -20 dB per decade (without C/O network) to -40 dB per decade (with 2^{nd} order network) and -60 db per decade (with a 3^{rd} order crossover network).

In the case of the vented box, it is important to notice that the sound pressure level that we hear is the result of adding the SPL produced by the diaphragm and the SPL produced by the air in the port. In the plot showing the SPL of the vented box system, it is easy to notice that at some frequencies, the SPL produced by the air in the vent is considerably larger that the SPL produced by the diaphragm.

Appendix

SPICE Circuits and Netlists

Infinite Baffle Circuits



Electrical



Mechanical



Acoustical

Closed Box Circuits



Electrical



Mechanical



Acoustical



Electrical With Matching Network



Electrical With Matching Network
and 2nd Order Crossover Network



Electrical With Matching Network
and 3rd Order Crossover Network

Vented Box Circuits



Electrical



Mechanical


U'p
Up
M'AP
MAB
Swuw
Uw
U'w
UL
Uo
- pw +
kpU'w
RA1P
RA2P
CA1P
MA1P
RAL
CAB
V3w
RA1w
RA2w
CA1w
MA1w
kwU'p
V4w
V5w
V7w
V6w

Acoustical

```
*------------------------------------------------
*Infinite Baffle
*------------------------------------------------
*INFINITE BAFFLE NETLIST
*DISPLAY VM(1)/IM(VD1) FOR INPUT IMPEDANCE
*DISPLAY 20*LOG10(VM(13)) FOR ON-AXIS PRESSURE
*DISPLAY VM(14) FOR DIAPHRAGM DISPLACEMENT

*ELECTRICAL CIRCUIT
VEG 1 0 AC 1V
RE 1 2 5.33
*LOSSY VOICE-COIL INDUCTANCE
GZE 2 3 LAPLACE {V(2,3)} = {1/(0.03*PWR(S,0.65))}
HBLUD 3 4 VD2 9.869
VD1 4 0 AC 0V

*MECHANICAL CIRCUIT
LMMD 5 6  0.025
RMS 6 7 2.544
CMS 7 8 7.584E-4
HBLIC 5 0 VD1 9.869
ESDPD 8 9 10 0 0.045
VD2 9 0 AC 0V

*ACOUSTICAL CIRCUIT
LMA1 10 12 5.314
RA1 10 11 7.938E3
RA2 11 12 17.998E3
CA1 10 11 3.654E-8
VD3 12 0 AC 0V
FSDUD 0 10 VD2 0.045

*ON-AXIS PRESSURE SOURCE
EPD 13 0 LAPLACE {I(VD3)} = {9390*S}
*DIAPHRAGM DISPLACEMENT SOURCE
EXD 14 0 LAPLACE {I(VD3)} = {1/S}


.AC DEC 77 10 100K
.PROBE
.END


*-------------------------------------------------
*CLOSED-BOX
*-------------------------------------------------
*DISPLAY VM(1)/IM(VD1) FOR INPUT IMPEDANCE
*DISPLAY 20*LOG10(16) FOR ON-AXIS PRESSURE
*DISPLAY VM(17) FOR DIAPHRAGM DISPLACEMENT

*ELECTRICAL CIRCUIT
VEG 1 0 AC 1V
RE 1 2 5.33
*LOSSY VOICE-COIL INDUCTANCE
GZE 2 3 LAPLACE {V(2,3)}={1/(0.03*PWR(S,0.65))}
HBLUD 3 4 VD2 9.869
VD1 4 0 AC 0

*MECHANICAL CIRCUIT
HBLI 5 0 VD1 9.869
LMMD 5 6 0.025
RMS 6 7 2.544
CMS 7 8 7.584E-4
```

```
ESDPD 8 9 10 13 0.045
VD2 9 0 AC 0

*ACOUSTICAL CIRCUIT
FSDUD 13 10 VD2 0.045
LMA1 10 12 2.657
RA1 10 11 3.969E3
RA2 11 12 9E3
CA1 10 11 7.308E-8
VD3 12 0 AC 0
LMAB 13 14 2.657
RAB 14 15 1.64E3
CAB 15 0 3.304E-7
RAL 15 0 4.818E6

*ON-AXIS PRESSURE DISPLAYS IN PROBE WITH 20*LOG10(VM(16))
EXP 16 0 LAPLACE {I(VD3)} = {59E3*S}
*DIAPHRAGM DISPLACEMENT DISPLAYS IN PROBE WITH VM(17)
EXD 17 0 LAPLACE {I(VD2)}={1/S}
.AC DEC 50 10 10K
.PROBE
.END


*---------------------------------------------------
*CLOSED-BOX WITH MATCHING NETWORK
*---------------------------------------------------
*DISPLAY VM(1)/IM(VD1) FOR INPUT IMPEDANCE
*DISPLAY 20*LOG10(16) FOR ON-AXIS PRESSURE
*DISPLAY VM(17) FOR DIAPHRAGM DISPLACEMENT

*ELECTRICAL CIRCUIT
VEG 1 0 AC 1V
RE 1 2 5.33
*LOSSY VOICE-COIL INDUCTANCE
GZE 2 3 LAPLACE {V(2,3)}={1/(0.03*PWR(S,0.65))}
HBLUD 3 4 VD2 9.869
VD1 4 0 AC 0
*-------------------------
*Matching Network
RR1 1 18 5.33
CC1 18 0 2.054E-5
RR2 18 19 2.484
CC2 19 0 1.293E-5
RR3 1 20 6.309
LL1 20 21 0.01
CC3 21 0 4.563E-4
*---------------------------

*MECHANICAL CIRCUIT
HBLI 5 0 VD1 9.869
LMMD 5 6 0.025
RMS 6 7 2.544
CMS 7 8 7.584E-4
ESDPD 8 9 10 13 0.045
VD2 9 0 AC 0

*ACOUSTICAL CIRCUIT
FSDUD 13 10 VD2 0.045
LMA1 10 12 2.657
RA1 10 11 3.969E3
RA2 11 12 9E3
CA1 10 11 7.308E-8
```

```
VD3 12 0 AC 0
LMAB 13 14 2.657
RAB 14 15 1.64E3
CAB 15 0 3.304E-7
RAL 15 0 4.818E6

*ON-AXIS PRESSURE DISPLAYS IN PROBE WITH 20*LOG10(VM(16))
EXP 16 0 LAPLACE {I(VD3)} = {59E3*S}
*DIAPHRAGM DISPLACEMENT DISPLAYS IN PROBE WITH VM(17)
EXD 17 0 LAPLACE {I(VD2)}={1/S}
.AC DEC 50 10 10K
.PROBE
.END

*------------------------------------------------------------
*CLOSED-BOX WITH MATCHING NETWORK AND 2ND ORDER C/O NETWORK
*------------------------------------------------------------
*DISPLAY VM(1)/IM(VD1) FOR INPUT IMPEDANCE
*DISPLAY 20*LOG10(16) FOR ON-AXIS PRESSURE
*DISPLAY VM(17) FOR DIAPHRAGM DISPLACEMENT

*ELECTRICAL CIRCUIT
VEG 1 0 AC 1V
*---------------------------
*2nd Order C/O Network
LLw 1 2 2.121E-3
CCw 2 0 1.866E-5
*----------------------------
*Matching Network
RR1 2 3 5.33
CC1 3 0 2.054E-5
RR2 3 4 2.484
CC2 4 0 1.293E-5
RR3 2 18 6.309
LL1 18 19 0.01
CC3 19 0 4.563E-4
*-----------------------------
RE 2 20 5.33
*LOSSY VOICE-COIL INDUCTANCE
GZE 20 21 LAPLACE {V(20,21)}={1/(0.03*PWR(S,0.65))}
HBLUD 21 22 VD2 9.869
VD1 22 0 AC 0
*---------------------------


*MECHANICAL CIRCUIT
HBLI 5 0 VD1 9.869
LMMD 5 6 0.025
RMS 6 7 2.544
CMS 7 8 7.584E-4
ESDPD 8 9 10 13 0.045
VD2 9 0 AC 0

*ACOUSTICAL CIRCUIT
FSDUD 13 10 VD2 0.045
LMA1 10 12 2.657
RA1 10 11 3.969E3
RA2 11 12 9E3
CA1 10 11 7.308E-8
VD3 12 0 AC 0
LMAB 13 14 2.657
RAB 14 15 1.64E3
CAB 15 0 3.304E-7
```

```
RAL 15 0 4.818E6

*ON-AXIS PRESSURE DISPLAYS IN PROBE WITH 20*LOG10(VM(16))
EXP 16 0 LAPLACE {I(VD3)} = {59E3*S}
*DIAPHRAGM DISPLACEMENT DISPLAYS IN PROBE WITH VM(17)
EXD 17 0 LAPLACE {I(VD2)}={1/S}
.AC DEC 50 10 10K
.PROBE
.END


*----------------------------------------------------------
*CLOSED-BOX WITH MATCHING NETWORK AND 3RD ORDER C/O NETWORK
*----------------------------------------------------------
*DISPLAY VM(1)/IM(VD1) FOR INPUT IMPEDANCE
*DISPLAY 20*LOG10(16) FOR ON-AXIS PRESSURE
*DISPLAY VM(17) FOR DIAPHRAGM DISPLACEMENT

*ELECTRICAL CIRCUIT
VEG 1 0 AC 1V
*--------------------------
*3rd Order C/O Network
LLco1 1 2 1.591E-3
CCco 2 0 4.977E-5
LLco2 2 3 5.302E-4
*---------------------------
*Matching Network
RR1 3 4 5.33
CC1 4 0 2.054E-5
RR2 4 18 2.484
CC2 18 0 1.293E-5
RR3 3 19 6.309
LL1 19 20 0.01
CC3 20 0 4.563E-4
*----------------------------
RE 3 21 5.33
*LOSSY VOICE-COIL INDUCTANCE
GZE 21 22 LAPLACE {V(21,22)}={1/(0.03*PWR(S,0.65))}
HBLUD 22 23 VD2 9.869
VD1 23 0 AC 0
*--------------------------


*MECHANICAL CIRCUIT
HBLI 5 0 VD1 9.869
LMMD 5 6 0.025
RMS 6 7 2.544
CMS 7 8 7.584E-4
ESDPD 8 9 10 13 0.045
VD2 9 0 AC 0

*ACOUSTICAL CIRCUIT
FSDUD 13 10 VD2 0.045
LMA1 10 12 2.657
RA1 10 11 3.969E3
RA2 11 12 9E3
CA1 10 11 7.308E-8
VD3 12 0 AC 0
LMAB 13 14 2.657
RAB 14 15 1.64E3
CAB 15 0 3.304E-7
RAL 15 0 4.818E6
```

```
*ON-AXIS PRESSURE DISPLAYS IN PROBE WITH 20*LOG10(VM(16))
EXP 16 0 LAPLACE {I(VD3)} = {59E3*S}
*DIAPHRAGM DISPLACEMENT DISPLAYS IN PROBE WITH VM(17)
EXD 17 0 LAPLACE {I(VD2)}={1/S}
.AC DEC 50 10 10K
.PROBE
.END


*-------------------------------------------------------
*VENTED-BOX
*-------------------------------------------------------
*DISPLAY VM(1)/IM(VD1) FOR INPUT IMPEDANCE
*DISPLAY VM(21) FOR DIAPHRAGM DISPLACEMENT
*DISPLAY VM(22) FOR PORT DISPLACEMENT
*DISPLAY 20*LOG10(VM(23)) FOR ON-AXIS PRESSURE
*DISPLAY 20*LOG10(VM(24)) FOR DIAPHRAGM PRESSURE
*DISPLAY 20*LOG10(VM(25)) FOR PORT PRESSURE

*ELECTRICAL CIRCUIT

VEG 1 0 AC 1V
REW 1 2 5.33
*LOSSY VOICE-COIL INDUCTANCE
GRA 2 3 LAPLACE {V(2,3)}={1/(0.03*PWR(S,0.65))}
HBLUW 3 4 V2W 9.869
V1W 4 0 AC 0V

*Mechanical Ckt

HBLIW 5 0 V1W 9.869
LMMDW 5 6 0.025
RMSW 6 7 2.544
CMSW 7 8 7.584E-4
ESDPW 8 9 17 10 0.045
V2W 9 0 AC 0V

*Acoustical CKT

FSDUW 10 17 V2W 0.045
LMABW 10 11 2.035
CABW 11 12 3.304E-7
V3W 0 12 AC 0
RALW 11 0 9.221E4
LMAP 11 13 49.355
FKP 15 13 V6W 0.097
LMA1P 13 15 7.971
RA1P 13 14 3.572E4
CA1P 13 14 2.707E-9
RA2P 14 16 8.099E4
V4W 15 16 AC 0V
V5W 16 0 AC 0V
FKWUP 19 17 V4W 0.29
LMA1W 17 19 2.657
RA1W 17 18 3.969E3
CA1W 17 18 7.308E-8
RA2W 18 20 9E3
V6W 19 20 AC 0V
V7W 20 0 AC 0V


*DIAPHRAGM DISPLACEMENT SOURCE
EXD 21 0 LAPLACE {I(V2W)}={1/S}
```

```
*PORT DISPLACEMENT SOURCE
EXP 22 0 LAPLACE {I(V5W)}={1/(5.027E-3*S)}
*ON-AXIS PRESSURE SOURCE
EPSUM 23 0 LAPLACE {I(V3W)}={9390*S}
*DIAPHRAGM PRESSURE SOURCE
EPD 24 0 LAPLACE {I(V7W)}={9390*S}
*VENT PRESSURE SOURCE
EPV 25 0 LAPLACE {I(V5W)}={9390*S}
.AC DEC 100 10 10K
.PROBE
.END
```